

回帰分析入門

1) 2 変量データの記述

1. 散布図の描画

【 課題 17 】 下に示したものは、日本の実質家計可処分所得と実質家計最終消費支出のデータ (平成 7 年基準、単位: 兆円)¹である。このデータを入力し、散布図を描いてみよう。

📖 散布図は次のような手順で描けばよい。

B2:C19 を範囲指定し、リボン内にグラフのグループにある、散布図 のボタンをクリックする。(挿入タブをクリックすることで表示される。)

散布図グラフのフォーマット(型式)メニューにおいて、「散布図(マーカーのみ)」(左上)をクリックし、リボンの中のレイアウト 1(左端)をクリックし、タイトルや軸ラベルなどが書き込めるようにする。


下の図のようにタイトル、軸ラベルを入力し、目盛線、凡例を非表示にする。

グラフの作成を終えた後で、それぞれの軸の書式設定をおこない、

縦軸 最小値：200 最大値：300 目盛間隔：20

横軸 最小値：240 最大値：340 目盛間隔：20

とする。



¹ 出典：『平成 17 年版 国民経済計算年報』

2. 相関係数の導出

【課題 18】 所得と消費のデータについて相関係数を求めてみよう。

相関係数の計算式は次のような式である。

$$R = \frac{n \sum XY - (\sum X)(\sum Y)}{\sqrt{\{n \sum X^2 - (\sum X)^2\} \{n \sum Y^2 - (\sum Y)^2\}}}$$

したがって相関係数を導出するためには、 X 、 Y 、 XY 、 X^2 、 Y^2 をまず求める必要がある。 X 、 Y は、B 列、C 列の和を求めればよいが、 XY 、 X^2 、 Y^2 を求めるためには、交差積(XY)と 2 乗(X^2 , Y^2)を D 列、E 列、F 列に計算したうえで、その和を求めることになる。

手順は次のようになる。

📖 手順

D 列に X と Y の交差積を求める。D2 セルに=B2*C2 と入力し、これをコピーすればよい。

E 列に X の 2 乗を、F 列に Y の 2 乗を求める。2 乗を表す演算子は '^' であり、E2 セルに=B2^2 と入力し、これをコピーする。F 列も同様である。

B21 セルから F21 セルに各列の合計を求める。これらのセルがそれぞれ X 、 Y 、 XY 、 X^2 、 Y^2 である。

C23 セルに =(18 * D21 - B21 * C21) / SQRT((18 * E21 - B21^2) * (18 * F21 - C21^2)) と入力する。この式と計算式とを見比べてみよ。

< 作成見本 >

	A	B	C	D	E	F
1	年	所得(X)	消費(Y)	XY	X^2	Y^2
2	1986	252	204	51408	63504	41616
3	1987	257	212	54484	66049	44944
4	1988	271	222	60162	73441	49284
5	1989	284	233	66172	80656	54289
6	1990	298	244	72712	88804	59536
7	1991	311	251	78061	96721	63001
8	1992	317	257	81469	100489	66049
9	1993	319	260	82940	101761	67600
10	1994	324	267	86508	104976	71289
11	1995	327	272	88944	106929	73984
12	1996	328	279	91512	107584	77841
13	1997	332	281	93292	110224	78961
14	1998	335	280	93800	112225	78400
15	1999	334	280	93520	111556	78400
16	2000	334	284	94856	111556	80656
17	2001	330	288	95040	108900	82944
18	2002	334	290	96860	111556	84100
19	2003	339	293	99327	114921	85849
20						
21	合計	5626	4697	1481067	1771852	1238743
22						
23		相関係数	0.980669			
24						

2) 単回帰モデル(その1)

1. 回帰直線の導出

【課題 19】所得と消費のデータについて $Y = \quad + \quad X$ という1次式をあてはめ、回帰係数、 \quad の推定値を求めよ。(C24セルに \quad の推定値 b 、C25セルに \quad の推定値 a を求めよ。)

回帰係数の推定値を求める式は次のようなものである。

$$b = \frac{n \sum XY - \sum X \sum Y}{n \sum X^2 - (\sum X)^2}$$
$$a = \frac{\sum X^2 \sum Y - \sum X \sum XY}{n \sum X^2 - (\sum X)^2}$$

この式に相関係数の導出の際に求めた $\sum X$ 、 $\sum Y$ 、 $\sum XY$ 、 $\sum X^2$ 、 $\sum Y^2$ を代入すれば回帰係数の推定値がそれぞれ計算できる。

2. 予測値と残差の計算

【課題 20】所得と消費のデータについて各年の X のデータに対する予測値 \hat{Y} と残差を求めよ。

📖 手順

G列に予測値を求める。予測値 \hat{Y} は各 X_i について $a + b X_i$ を計算すればよいので、G2セルに1986年の X (B2セル) に対応する予測値を求めるなら $=\$C\$25 + \$C\$24 * B2$ とし、これをコピーすればよい。ここでは、コピーの際に絶対参照をするので、'\$'がついている。

H列に残差を求める。残差は Y から予測値 \hat{Y} を引いたものなので、H2セルに $=C2 - G2$ とし、これをコピーすればよい。

3. 回帰直線のグラフへの書き入れ

散布図に回帰直線を書き入れる場合、Excelでは各 X に対応する予測値をグラフに書き入れ、それを直線でつなぐという手順をとる。

【課題 21】所得と消費のデータについて散布図に回帰直線を書き入れよ。

📖 手順

グラフをアクティブにした状態で、リボンの中の「データの選択」ボタン(出ていない場合には上部の「グラフツール」をクリックする)をクリックし、凡例項目(系列)の「追加」ボタンをクリックする。そして、「系列 X の値」を **B2:B19**、「系列 Y の値」を **G2:G19** とする。



の操作で散布図上に赤色のマーカーが現れたはずである。これを直線で結ぶ。赤色のマーカーのひとつを右クリックして、「データ系列の書式設定」を選ぶ。そこで「マーカーのオプション」をクリックし、マーカーの種類として「なし」をチェックし、「線の色」をクリックし、線の色として「線(単色)」をチェックする。グラフエリアの外をクリックすると回帰直線が引けたことがわかるはずである。

< 作成見本 >

日本の実質家計可処分所得と実質家計消費支出 E37-000 徳山 太郎

年	所得(X)	消費(Y)	XY	X^2	Y^2	予測値	残差
1986	252	204	51408	63504	41616	202.2889	1.711132
1987	257	212	54484	66049	44944	207.132	4.868011
1988	271	222	60162	73441	49284	220.6927	1.307272
1989	284	233	66172	80656	54289	233.2848	-0.28484
1990	298	244	72712	88804	59536	246.8456	-2.84558
1991	311	251	78061	96721	63001	259.4377	-8.4377
1992	317	257	81469	100489	66049	265.2494	-8.24944
1993	319	260	82940	101761	67600	267.1867	-7.18669
1994	324	267	86508	104976	71289	272.0298	-5.02981
1995	327	272	88944	106929	73984	274.9357	-2.93568
1996	328	279	91512	107584	77841	275.9043	3.095693
1997	332	281	93292	110224	78961	279.7788	1.221196
1998	335	280	93800	112225	78400	282.6847	-2.68468
1999	334	280	93520	111556	78400	281.7161	-1.71605
2000	334	284	94856	111556	80656	281.7161	2.283948
2001	330	288	95040	108900	82944	277.8416	10.15844
2002	334	290	96860	111556	84100	281.7161	8.283948
2003	339	293	99327	114921	85849	286.5592	6.440827
合計	5626	4697	1481067	1771852	1238743		

相関係数 0.980669
 b 0.968624
 a -41.8044
 決定係数 0.961712



3) 単回帰モデル(その2)

前章では、相関係数と回帰係数の推定値を交差積和($\sum XY$)、2乗和($\sum X^2$, $\sum Y^2$)を求め、それを計算式に代入することによって求めた。

しかしExcelによって相関係数と回帰直線を求めるには、以下に説明するような Excelが備えている関数を用いることもできる。ここではregの例について、統計関数を用いた方法についても行ってみたいことにする。その際には、データを入力したセルの範囲に名前をつけておくと便利である。まず、実習の準備として、regのSheet1から、年次、所得X、消費Yの部分(A3:C21)をSheet2のA3:C21にコピーし、所得のデータに $_X$ 、消費のデータに $_Y$ という名前を定義しておく。

1. 統計関数による相関係数と回帰直線の導出

Excelが備えている関数を用いた相関係数と回帰直線の導出を行ってみることにする。regのSheet2のA3:C21に、年次、所得X、消費Yのデータが入力されているものとする。

(1) 関数 PEARSON (CORREL) , RSQ

相関係数を求めるには、関数 **PEARSON**(引数1, 引数2)を用いる。PEARSON は相関係数を最初に導出した Karl Pearson (イギリス;1851-1936) にちなんでつけられた名前である。または、**CORREL** という名前の関数もあるが、どちらも全く同じものである。引数は2個あり、それぞれがデータの範囲(名前でもよい)である。戻り値は r_{xy} である。

```
= PEARSON( B4:B21, C4:C21 )  
= PEARSON(  $\_X$  ,  $\_Y$  )
```

範囲B4:B21に名前 $_X$ 、範囲C4:C21に名前 $_Y$ を付けてあれば、どちらの式でも結果は同じである。以下の説明では下式の書き方で示す。

関数 **RSQ**は相関係数の2乗(= **決定係数**)を求める関数であるが、引数は PEARSON と同じである。したがって、べき乗を求める演算子 ^ を用いれば RSQ は不要となる。

```
= RSQ(  $\_X$  ,  $\_Y$  )  
= ( PEARSON(  $\_X$  ,  $\_Y$  ) ) ^ 2
```

どちらも全く同じ結果を与える。

(2) 関数 SLOPE とINTERCEPT²

SLOPE は回帰直線の傾き(回帰係数) b を、**INTERCEPT** は切片(回帰定数) a を求める関数で、どちらも引数は2個あるが、最初の引数が従属変数の範囲で、2個目の引数が独立変数の範囲をとる。引数の順序に注意しなければならない。

```
= SLOPE(  $\_Y$  ,  $\_X$  )  
= INTERCEPT(  $\_Y$  ,  $\_X$  )
```

² 回帰直線の傾きと切片を求める関数には、LINEST という関数がある。この関数は傾きと切片以外に分析結果に関する多くの情報量を与えてくれる、非常に便利な関数である反面、使用法および結果の解釈の仕方が難しい。LINEST 関数の説明はここでは省略する。

(3) 関数 FORECAST と TREND

予測値 \hat{Y} を求める関数には2種類のものを用意されている。関数**FORECAST** は引数を3個とり、FORECAST(X_i , Y範囲, X範囲) として用いる。戻り値は $a + bx_i$ として求められた数値1個である。

=FORECAST(B4, _Y, _X) セルB4の値を x としたときの $a + bx$ が戻り値

残りの X の値に対する予測値は、これをコピーして求めればよい。

あるいは、n個の予測値を書き込む範囲を指定しておき、配列数式とすることもできる。たとえば、D4:D21の範囲を指定して、

=FORECAST(_X,_Y,_X)

を入力して、**Ctrl** + **Shift** + **Enter** とする。

関数 **TREND** も同じ予測値を求めるものであるが、引数の数が FORECAST より1個多く、計4個となる。一般的な型式は TREND(Y範囲, X範囲, X_i, 1)となる。最後の引数は 0 か 1 で、0 のときは、原点を通る直線 $Y = bX$ による予測値、1 のときはこれまで通りの $Y = a + bX$ による予測値を戻り値として求める。第4引数を省略した場合は、1 を指定したものとみなす。

TRENDの第3引数として、a, b の計算に用いない任意の数値を指定することもできる。たとえば、

=TREND(_Y, _X, 190, 1)

とすれば、 $\hat{Y} = a + b \times 190$ を求めることになる。regの X のデータの中には 190という数値はなく、これによって求まる \hat{Y} は未知の X の値に対する予測値(外挿値)である。同様のことを、FORECASTを用いてもおこなうことができる。

=FORECAST(190, _Y, _X)

FORECASTでは 190 が第1引数となる点に注意されたい。また、190 という数値を直接指定するのではなく、セル番地で指定することもできる。セル N4 に190が書き込んであれば、

=TREND(_Y, _X, N4, 1)

=FORECAST(N4, _Y, _X)

とすればよい。また、N4からN13に $\hat{Y} = a + b x$ として求めたい x の値が連続して書き込まれていれば、

=TREND(_Y, _X, N4:N13, 1)

=FORECAST(N4:N13, _Y, _X)

とすればよい。範囲 N4:N13 に名前を付けて、それを使用してもよい。

2. 分析ツールの利用

Excel には統計分析を行うためのいくつかの分析ツールが付属している。これらのツールを使えば一度に詳細な分析結果を得ることができる。

分析ツールを最初に使用する場合には、アドイン(有効にすること)しなくてはならない。分析ツールのアドインは次のようにおこなう。

左上の **Office ボタン** を押し、下にある「Excel のオプション」のボタンをクリックする。

「分析ツール」にチェックをつけ、OK ボタンをクリックする。

すると、データタブの中に「データ分析」のボタンが出てくるので、下のほうにある、回帰分析を選べばよい。

アドインを行った後で、再びメニューバーから「ツール」を選ぶと、下のほうに「分析ツール」と表示される。ここで分析ツールを選び、回帰分析を選べばよい。


4) 単回帰モデルの他の実例

1. フィリップス曲線

フィリップス(A. W. Phillips) は 1958 年、名目賃金上昇率と失業率の間に右図のような関係があることを、1861 年から 1957 年のイギリスのデータを用いて示した。この曲線は、発見者の名前をとって**フィリップス曲線**と呼ばれる。

その後の研究により、賃金上昇率は物価上昇率との関係が強いことなどから、最近では縦軸に物価上昇率を用いることが多い。

この関係は、インフレ率が高い状況では失業率が低下し、失業率が高い状況ではインフレ率が低下するという、インフレーションと失業とのトレードオフの関係を示している。



2. フィリップス曲線の定式化

フィリップス曲線は、物価上昇率を Y 、失業率を X とするとき、

$$Y = \alpha + \beta \frac{1}{X}$$

という式をあてはめることが考えられる。この式において、 Y と X の関係は直線では表されない(Y と X は線形関係ではないという)が、 $X' = \frac{1}{X}$ という変数変換をおこなえば、この式は

$$Y = \alpha + \beta X'$$

という直線(線形式)で表すことができる。

3. フィリップス曲線の実証分析

近年の日本のデータにおいて、フィリップス曲線のような関係が成り立っているかどうかを分析してみよう。

【課題 24】講義用 HP に **Phillips.xls** という名前で 1985 年から 2007 年までの物価上昇率と失業率のデータ³が保存されている。このデータを各自のフォルダにダウンロードし、分析をおこなってみよう。

手順は次のようになる。

📖 手順

失業率を横軸に、物価上昇率を縦軸にとった散布図を描く

D 列に $X' = \frac{1}{X}$ という変換をおこなったデータを作成する。D2 セルに '=1/C2' と入力し、

それを D 列全体にコピーすればよい。

分析ツールを用いて Y を被説明変数、 X' を説明変数とする回帰分析をおこなう。

分析ツールの出力結果にある予測値を、散布図に描き入れ、線でつなぐ。

³ 物価上昇率 - 総務省統計局『消費者物価指数』(全国・総合)対前年比
失業率 - 総務省統計局『労働力調査』完全失業率(年平均・男女計)